
 

 

 

er 
 

 

 
 

 

 
 

Sinclair QL retro  gaming 

 

 

 

 

 

 

 

 

 

 
 

Sinclair QL retro gaming 

 



 

 

  

 

 

 
 

 

 
 

 
 

 

 
 

 

 
 

 

 
 

 

 
 

 

 
 

 

 
 

 



 

Page 1 

 

 

 

 
 

 

 
QBITS Galaxy AD2370DL  
As to my thoughts behind the QBITS Galaxy AD2370, the early to mid nineteen eighties 

was a time when I had perhaps an overly enthusiastic interest in Science Fiction. The first 

of the Star Wars films was released and my family were considered by some to have an 

addiction to playing board games, which included variants of the game RISK. I guess all 

played their part in how this Game idea originated and began its development from early 

concept to a working program. 

 

QBITS Galaxy AD2370 - Basic Concept 

My to do list began:- create a QL Computer Game where Starships battle it out with laser 

beam weapons  to capture a Galaxyôs Star Systems. To expand upon this there was to be 

the ability to exchange Technology and Trade thereby raising Populations as a necessity 

to Recruiting more Troops. Multiple players were a consideration, or perhaps just the 

Computer to imitate the moves and actions of a human player. The winner of the game 

was the one who had captured all of their opponents Star Systems, leaving them unable 

to recover and retaliate. The stratagems deployed to achieve a WIN ultimately to depend 

on the Roll of the Dice, where the final result can never be certain. 
 

Create an Intro page with basic instructions on how to play the Game before moving to a 

Menu providing choice of side plus the normally expected options of New, Load, Save 

and Exit. Levels of difficulty was added later to the list. 
 

The background for the Game was as expected to be the Star Systems of a Galaxy far, far 

away (No surprise there). Early beginnings, which have persisted was with just two sides 

the Rebel Alliance and Federation of Planets. At initialisation some of the Star Systems 

randomly allocated to either side, others left as independent.  
 

Each Star system to be given various levels of Population, Technological development 

and Interstellar Trading abilities. Other parameters covered such things as Galaxy Map 

coordinates, a Planetôs surface colour, then thoughts on alternative directions when 

approaching the Planet within a Star System etc.    
 

Later additions were the End Game sequence, a Final Battle between the Rebel Alliances 

defending their home Planet against the Federations Death Star. Then to follow the Final 

Victory a high score League Table, which can be amended with the name, nickname or 

pseudonym of the new high scorer.    

 
Quick Note:  I developed most of this code using the QL2K emulator on a six years old 

Desktop running Win 7. The speed is mega fast than even a compiled program running on 

an original QL hardware. Therefore, running QBITS Galaxy AD2370 at anything of a 

respectful speed you are going to need expanded memory and a later much faster version 

of QL Hardware or download and use an emulator such as QL2K or SMQL on a more 

modern computer (See page 36 for downloads). 

QBITS Galaxy AD2370 - The Beginnings 

 

Initial thoughts began with constructing 

a Galaxy with Stars Systems in the form 

of a Territorial Map.  

 
 

 

 

   

 

 

 

 
This changed to a Galaxy Map of random POINTs (Stars) with Grid Position of the 

selected Star identified by a double circle and the crossover point of horizontal & vertical 

Grid Lines that could be moved from within a REPeat loop by use of the cursor keys. 

 

QBITS Development - Galaxy AD2370 Space View Screen 

An action view screen I guess is an expectation with any Computer Game. For Galaxy 

AD2370 this had to be moving through the Stars of space. To create this ill usion the view 

screen was subdivided into quarters (four sub-windows), then using the PAN and 

SCROLL commands randomly placed POINTs (Stars) are moved directionally out from 

the centre of the view screen towards the four corners.  

 

After playing around with the number of 

generated Star POINTs and the SCROLL 

and PAN values the code below gave a 

respectable early result.   

 
100 REMark QBGT01 (Test for Space View) 
101 : 
102 WINDOW 512,256,0,0:PAPER 0:CLS 
103 DATA 304,152,98,19 
104 DATA 150,75,100,20,150,75,250,20 
105 DATA 150,75,100,95,150,75,250,95 
106 RESTORE 103 
107 : 
108 FOR i=3 TO 7:OPEN#i,scr_:READ a,b,c,d:WINDOW#i,a,b,c,d 
109 BORDER#3,1,2:INK#3,7 
110 FOR i=1 TO 100:POINT#3,RND(10 TO 180),RND(10 TO 90) 
111 REPeat lp 
112   FOR i=1 TO 6:POINT#3,RND(20 TO 180),RND(10 TO 90) 
113   PAN#4,-2:SCROLL#4,-2:PAN#5,2:SCROLL#5,-2 
114   PAN#6,-2:SCROLL#6,2:PAN#7,2:SCROLL#7,2 
115   PAUSE 5 
116   IF KEYROW(1)=8:FOR i=3 TO 7:CLOSE#i:STOP 
117 END REPeat lp 

QBITS Development - Galaxy AD2370 Planetary Orbit 

In exploring ways to emulate approaching a Star System and then coming into orbit 

above a Planet, both scenarios were accomplished by the redrawing an enlarging disk in 

the form of a SuperBASIC FILLed CIRCLE.  
 

Galaxy Map, Space view and Planetary Orbit are all combined in the test program below. 
 
100 REMark QBGT03 (Test for Galaxy Map - Space View - Orbit) 
 

104 WINDOW 512,256,0,0:PAPER 0:CLS 
106 DATA 304,152,98,19 
108 DATA 150,75,100,20,150,75,250,20 
110 DATA 150,75,100,95,150,75,250,95 
112 DATA 304,12,98,2 
 

116 sx=164:sy=48:RESTORE 106:AT#2,18,0:PRINT#2,ôQBG03ô 
 

120 InitWin:InitGrid:Main 
 

124 DEFine PROCedure InitWin 
126 FOR i=3 TO 8:OPEN#i,scr_:READ a,b,c,d:WINDOW#i,a,b,c,d 
128 BORDER#3,1,2:INK#3,7:BORDER#8,1,2:INK#8,7 
130 FOR i=1 TO 100:POINT#3,RND(10 TO 160),RND(10 TO 90) 
132 END DEFine 
 

Creating the Galaxy Map with its Stars and Grid was not difficult, moving the Map Grid 

Lines the XOR function of OVER is used.  
 
 

136 DEFine PROCedure InitGrid 
138 OPEN#13, scr_:WINDOW#13,260,60,120,170 
140 OPEN#14, scr_:WINDOW#14,90,22,8,6 
142 CLS#13:BORDER#13,1,7:INK#13,2:SCALE#13,100,0,0 
144 FOR h=12 TO 96 STEP 18 
146  LINE#13,6,h TO 326,h 
148 END FOR h 
150 FOR i=16 TO 330 STEP 20 
152  LINE#13,i,3 TO i,98 
154 END FOR i 
156 INK#13,2:FOR i=1 TO 60 :POINT#13,RND(60 TO 260),RND(30 TO 70) 
158 INK#13,4:FOR i=1 TO 30 :POINT#13,RND(80 TO 240),RND(40 TO 60) 
160 INK#13,7:FOR i=1 TO 180:POINT#13,RND(20 TO 310),RND(10 TO 90) 
162 OVER#13,1:PRINT#13,'   use the cursor keys to move grid line':OVER#13,0 
164 END DEFine 
 
 
168 DEFine PROCedure Gridpos 
170 IF sx< 10:sx= 10 
172 IF sx>320:sx=320 
174 IF sy<  5:sy=  5 
176 IF sy> 95:sy= 95 
178 OVER#13,-1:INK#13,4 
180 LINE#13,sx,5 TO sx,95:LINE#13,12,sy TO 320,sy:OVER#13,0  
182 CLS#14:INK#14,2:PRINT#14,'Grid Position'\' x:';sx;' y:';sy 
184 AT#2,18,0:PRINT#2,ôQBGT03ô 
184 END DEFine 

190 DEFine PROCedure Main 
192 CLS#8:PRINT#8,' Press any key to enter orbit  -  esc to quit' 
194 Gridpos 
196 REPeat lp 
198   StarMove 
200   IF KEYROW(1)=8:FOR i=3 TO 20:CLOSE#i:STOP 
202   IF KEYROW(1)=64 :StarView:Eorbit:PAUSE:Lorbit 
204   IF KEYROW(1)=  2 :IF sx>5   :Gridpos: sx=sx-5  :Gridpos 
206   IF KEYROW(1)=  4 :IF sy<95  :Gridpos: sy=sy+5  :Gridpos 
208   IF KEYROW(1)= 16 :IF sx<320 :Gridpos: sx=sx+10 :Gridpos 
210   IF KEYROW(1)=128 :IF sy>5   :Gridpos: sy=sy-10 :Gridpos 
212 END REPeat lp 
214 END DEFine 
 

218 DEFine PROCedure StarMove 
220  FOR i=1 TO 10:POINT#3,RND(10 TO 160),RND(10 TO 90) 
222  PAN#4,-2:PAN#5,2:PAN#6,-2:PAN#7,2 
224  SCROLL#4,-2:SCROLL#5,-2:SCROLL#6,2:SCROLL#7,2 
226  PAUSE 5 
228 END DEFine 
 

Note: This is the simple use of PAN and SCROLL although order and size of move are 

important to get the right balance... 
 
232 DEFine PROCedure StarView 
234 CLS#8:PRINT#8,' Approaching Alpha Centauri' 
236 FOR a=1 TO 8 
238   INK 0:FILL 1:CIRCLE 90,70,a*.6:FILL 0 
240   x=90:y=70:m=a*.6:Star 
242   FOR b=1 TO 6:POINT#3,RND(10 TO 160),RND(10 TO 90) 
244   PAN#4,-1:SCROLL#4,-.5 
246   PAN#6,-1:SCROLL#6,.5:PAN#7,1:SCROLL#7,.5 
248   PAUSE 10 
250 END FOR a 
252 END DEFine 
 

256 DEFine PROCedure Star 
258  INK 2  :FILL 1:CIRCLE x,y,m*1  :FILL 0 
260  INK 241:FILL 1:CIRCLE x,y,m*.95:FILL 0 
262  INK 7  :FILL 1:CIRCLE x,y,m*.9 :FILL 0 
265  INK 241:FILL 1:CIRCLE x-m*.6,y-m*.4,m*.2,.3,PI/4:FILL 0 
266 END DEFine 

 

 
The illusion of a rising Planet crest is achieved in 

SuperBASIC by drawing a large FILLed CIRCLE, 

where most of it graphics resides off screen. Moving 

the x y coordinates set off screen upwards and 

redrawing the circle gives the illusion of a rising crest.  

270 DEFine PROCedure Eorbit 
272 CLS#8:PRINT#8,' Entering Planetary Orbit' 
274 FOR i=1 TO 20 
276    SCROLL#3,-1:PAUSE 2 
278    POINT#3,RND(10 TO 60), RND(5 TO 95) 
280 END FOR i 
282 PAUSE 5:INK#3,241 
284 FOR p=-240 TO -190 STEP 2 
286   FILL#3,1:CIRCLE#3,76,p,200:FILL#3,0:SCROLL#3,-1:PAUSE 2 
288   POINT#3,RND(10 TO 160),RND(5 TO 10) 
290 END FOR p 
292 CLS#8:PRINT#8,' Press any key to leave Orbit' 
294 END DEFine 
 

298 DEFine PROCedure Lorbit 
300 CLS#8:PRINT#8,' Leaving Orbit' 
302 CURSOR#3,0,60 
304 FOR i=1 TO 10 
306   SCROLL#3,2,2 :POINT#3,RND(40 TO 120),RND(10 TO 90) 
308   SCROLL#3,-1,1:PAN#4,-2:SCROLL#4,-1:PAN#5,2:SCROLL#5,-1 
310   PAUSE 2 
312 END FOR i 
314 CLS#8:PRINT#8,' Press spacebar for orbit  -  esc to quit' 
316 END DEFine 
 

These early code investigations gave a starting point to develop further variations and 

later incorporate them as part of the full program.  

 
Note on Code Development  

I did not consciously consider before writing my Game Programs that I might work to a 

framework. However, something of such has evolved. Itôs not that complicated and 

seems good sense to use in identifying areas of work. 
 

The premise of the three-act play I use when writing fiction appears to lend itself to 

developing Game code. Act 1 the Welcome, covering the Intro page, Menu and 

Initialisation. Act 2 the Game, review of information, taking action, their result. Act 3 the 

Game End, the Winners recognition, and recording the results.  
 

(1) Welcome:-  an Intro  page to describe the basic game and its function. Initialis ation 

of Graphics and Parameters in line with opening choices made from a Menu. For Galaxy 

AD2370 this is a choice of Sides and Skill L evels, plus the (N)ew (L )oad (S)ave 

(H)ighscore (E)xit much of which was covered by earlier progs. 

 

(2) Challenge:-  subdivides into multiple small acts of Selection derived from review of  

information displayed, then taking the appropriate choice of Action in order to progress, 

followed by assessment of the Outcome successes or failures.   
 

(3) Game End:-  when all the challenges have been met, a sequence depicting the Final 

Battle between the opposing sides, a declared Winner and then some form of League 

Table showing high scores that can be updated when necessary. 

QBITS Galaxy AD2370 Intro                     PART1 

An Intro page aim is to give some insight to the Games parameters, role and goals. They 

can be just printed lines of text. However, there is no reason not to embellish it with a 

imaginative Title and background imagery to spice things up a little.   

 

QBITS Title     

QDOS SuperBASIC commands CSIZE with use of OVER printing to a CURSOR offset 

presents the opportunity to create 3D style character effects. Trying out different colour 

arrangements, can produce some pleasing even iconic results. 

 
128 WINDOW#1,512,256,0,0 :PAPER#1,0:CLS#1:CSIZE#1,2,1:OVER -1  
130 INK#1,2:FOR i=1 TO 3:CURSOR#1,140+i,20-i:PRINT#1,'QBITS Galaxy AD2370'  
132 INK#1,7:FOR i=1 TO 2:CURSOR#1,140+i,20-i:PRINT#1,'QBITS Galaxy AD2370'  
134 OVER 0:CSIZE 0,0:  

 

This code was where I began, experimenting with different INK and i Offset values.  

 
Developing the Opening Scene 

Artistic licence or not, I guess this is a matter of choice and limited only by the level of 

understanding and usage of QL graphics. There are two, possibly more directions in 

which to create a background. The simple and obvious approach is a static colour pattern, 

making sure the text printed over it is still legible. A more complex one will involve 

changing and/or moving images. If the Game has more than one static image or better 

still a moving graphic sequences, this presents an opportunity to consider them for use as 

part of the Game Intro.   

  

For Galaxy AD2370 I wanted to create an interesting opening. The Title to be followed 

by a two line statement, setting the scene and printed to screen character by character, all 

to give the illusion of someone typing an urgent message in real time.  

 

To create this affect I considered the INK colour, CURSOR position and a String length 

(number of characters). The string variable for Game message Gm$(2,70) is first 

dimensioned as multiple strings then each string of characters is loaded.  

 
110 DIM Gm$(2,70): RESTORE 122:FOR r=1 TO 2:READ Gm$(r) 
 

120 DATA 7,90,50,7,76,60 
122 DATA 'It is a turbulent time in the Galaxy as the Federation' 
124 DATA 'seeks to enforce new rules on Interstellar Travel and Trade.' 
 

128 RESTOR 120 
 

130 FOR a=1 TO 2 
132    READ i,c,r:INK i:CURSOR c,r:FOR b=1 TO 70:PAUSE 1:PRINT Gm$(a,b); 
134 END FOR a 

 

The FOR loop (a) reads the INK colour (i), CURSOR column (c) and row (r), setting the 

INK colour and CURSOR position. An embedded FOR loop (b) reads each string of 

characters and prints them one at a time to the screen. Use of the FOR loop is extended in 

the Galaxy AD2370 code to print further lines of the Intro. 

Here the graphic sequence using the 

components of the End of Game namely 

the DeathStar and Starships imagery.  

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

QBITS Galaxy Full Introduction  

This expresses some of the Games 

actions and requirements. 

  

 

 

 

 

 

 

 

 

 

QBITS Galaxy Game Menu  

The next step is the playerôs choice 

of Skill  and the Side to play on.  

QBITS Galaxy Load/Save  

Save and Load in Galaxy AD2370 are functions to store and retrieve Game Data so it can 

be continued later. In previous QBITS programs, these functions have been implemented 

in a variety of ways. The simplest is a single file, where the QL command PRINT writes 

to the file opened with OPEN_NEW for Save and the INPUT command reads from the 

opened file with OPEN_IN for Load.  

 
Save 
120 DELETE device_filename$     Delete Old File First  
121 OPEN_NEW#99,device_filename$ 
122 FOR a=1 TO 70 
124   FOR b=1 TO 9:PRINT#99,astro(a,b):END FOR b  Galaxy Star Info  
125 END FOR a 
126 FOR a=1 TO 2 
127   FOR b=1 TO 3:PRINT#99,side(a,b) :END FOR b  Galaxy Side Info 
128 END FOR a 
 

Load 
130 OPEN_IN#99,device_filename$ 
131 FOR a=1 TO 70 
132   FOR b=1 TO 9:INPUT#99,astro(a,b):END FOR b 
133 END FOR a 
134 FOR a=1 TO 2 
135   FOR b=1 TO 3:INPUT#99,side(a,b) :END FOR b 
136 END FOR a 

 
For multiple files a little more is required. QBITS Galaxy scrolls through a short list in 

the form of a specific filename with an appended number  QBAD2370DL_1 to 8. A more 

complex list with differing file names as well as the device used, mdv1_ flp1_ win1_ etc. 

can also be part of the selection as is the case provided with QBITS FTidy program.   

 

 

 

 

 

 

 

 

 
The QBITS default device is flp1_ < this can be changed to users choice > see line 104 of 

QBITS Galaxy AD 2370 program 

 

While working on the Skill levels and their time delays the idea of a League Table arose. 

Keeping track of Game Time then became of paramount importance. Therefore it needed 

to be part of the Game info Saved and when Loaded become the starting point of the 

continued Game Time Session (GTS). 

QBITS Galaxy AD2370 Challenge       PART 2 

What enhances a Game is to be able to access information and take decisions, RISKôs to 

your outcome. Developing a stratagem to achieve success is the playerôs prerogative. For 

this to be valid, my first objective was one of displaying the status information of each 

Side, initialised at the start and then updated as the Game progressed.  

 

Star Systems 

The arrays astro$(70,20) stores a Star Systems name and astro(70,9) holds the details of 

each. This begins with Technological advance (Tec 1 to 12), Trading abilities (Trade 1 to 

20) and Machines of Destruction MODs (1 to 20) for defence. These entries are used in 

the exchange process of the Transfer Phase of Play.  
 

The next entry is for Population (0.50 to 10 Billion), a random number that is used in 

calculating the initial Tec/Trade credits and MODs. The fifth entry is ownership, again 

randomly set for each new game:- Independent=0, Rebel Alliance=1, Federation=2. The 

values of these entries change as the Game moves forward. 
 

The next four are the Galaxy Map x, y coordinates, Planet surface colour and direction of 

Orbital approach (These do not change during the Game).   

 

Opponents Status  

The array side(2,3) holds Tec/Trade and MOD entries for each side. For Attack each 

player needs a Starship (1 to 5) calculated from the total number of Starship based MODs 

(1 to 100) and for the Transfer Phase Tec (1 to 24) and Trade (1 to 40) credits. The 

updated screen displays for this also show the number of annexed Territories (1 to 70).  
 

The Game begins with 10 Territories (Star Systems) allocated to each side, the Rebel 

Alliance or Federation. Land based MODs are tallied up and any differences compared. 

If equal, 20 MODs are allocated to the inventory of each opponentôs Starships. If not the 

opponent with the least number of Land based MODs has the difference added to their 

20. Therefore, both sides start the Game with the same number of MODs.  

 

Selecting Star Systems  

Investigating several arrangements 

the final method  chosen was to 

simply  Scroll through the star list 

using the Up/Down cursor keys 

and making a selection with the 

Spacebar. As the Star Systems are 

scrolled through, the Galaxy Map 

position is highlighted by cross 

wires and a double circle.  
 

 

 

 

 

The Grid coordinates are displayed top left, the Star System Name  

and details on the right::- Population, Tec/Trade and MODs etc.    

Space Travel 

Space is vast, the average distance across a Galaxy is a 100,000 Light years. Hyperspace 

or Subspace travel is almost instantaneous, but needs accurate star charts and calculated 

navigations to plot a course avoiding any collision. Slower warp speeds can take several 

days, weeks or even months to reach a Star system. Travelling on Impulse (sub light 

speed) can still take several hours to cross a solar system. 

 

Hyperspace Jump, Warp Speeds & Impulse  

For the Galaxy AD2370 Game my premise was to depict the journeys of space travel 

with a graphic sequence. This became a Hyperspace Jump to the selected Star, followed 

by a rapid deceleration of Warp Speed on approaching the Star and then a swing into 

Planetary Orbit on Impulse drive.   

 
Attack Mode 

Upon arriving in orbit over a Star Systems Planet if not already annexed then Attack 

Mode is activated. The Roll of the Dice determines which side loses MODs, the outcome 

either a defeat for the defending Planet or a retreat into space by the Attackers Starship. If 

a Victory for the attacker, the Star System is annexed and calculations determined by the 

Planets Population lead to Tec/Trade and MODs being added to the Starship inventory. 

 

Transfer Mode 

If the Planet is already annexed or has been defeated, Transfer Mode is activated. Using 

the cursor keys, Transfer Tec/Trade credits to the Planet. Then use the (R)ecruit key to 

increase the Population and generate more MODs.  

 

Tec/Trade and MODs can then be Transferred back to the Starship to strengthen forces 

for future Attacks. Alternatively, leaving higher numbers of MODs in defence of your 

annexed Star System makes them less vulnerable to opponent attacks. 

 

Warning:-  Transfer Mode and (R)ecuit might be your only means of raising  

MODs following heavy attack losses.  

QBITS Galaxy AD2370 CPlay 

For the computer to act as a worthy opponent, combining stratagems and logic to create a 

workable code required a good amount of forethought. My programming skills being 

limited, no way was this going to approach anything like Artificial Intelligence.  
 

The obvious route to follow was the actions dictated by the Game, namely seek out a Star 

System, Attack and/or Transfer Tec/Trade credits, recalculate the Population and thereby 

generate and Recruit more MODs. However, if this is to be truly representative, then the 

same fortunes and misfortunes when Rolling the Dice must apply as to a human player.   

 

The Proôs and Conôs of Conflict 

At Game start if the computer is the one awarded the extra MODs to their Starship fleet, 

this is a straight advantage when sallying forth to Attack Star Systems. Yet the fickle 

hand of fate when Rolling the Dice can be both fortuitous and equally as quick lead to a 

pending disaster.  
 

To guard against the worst outcomes, avoidance mechanisms needed to be part of the 

Computer playerôs code. For example before an Attack, the number of Starship MODs 

should be maybe three or four times larger than the Star Systems defence MODs. As the 

Game progresses if the Computers Starships MODs fall low in value, the computer 

player should avoid attacking Star Systems and seek to recruit more MODs from 

Transfers with its own Star/Planet Systems. To a Human player these actions can be a 

relatively easy decision to make. For the Computer player however, what criteria does it 

use, does it involve analysing and weighing the Proôs and Conôs of each and every Star 

System against some master plan of action or keep it as a set of simple random choices.  

 

Computer Interaction 

How and when should the Computer interact?  I considered a number of options. The 

first and obvious is a straight flip-flop between Human and Computer. When working 

with long time delays, the Computer player reliant on random number choices, placed it 

at a distinct disadvantage. Potentially lesser actions of value would be accomplished 

leading to a relatively easy win by most human players.  

 

Alternatives   

Here my thoughts again were on letting the computer compete for access in the main 

Game loop by control of a random number. A short experimentation with this form of 

access depending on choice and range of random numbers, proved either ineffectual or at 

times allowed the computer to block any human opponents interaction leading to an 

impossible task of a human player ever winning.  

 

Variable Level 

In the end, I decided on a time delay that locked the Computer out of play rather than a 

straight flip-flop or reliance on a random period. By increasing or decreasing the period 

of delay, the time for the human player to complete their actions gave the guise of a 

variable Skill level.  

QBITS Galaxy AD2370 Game End      PART 3 

A few thoughts on winning, essentially the winnerôs opponent has to have lost all of their 

Star System Territories and Land based MODs. Although they may still have Starship 

based MODs and the slimmest of chances of a comeback, I decided to base a WIN on the 

losing opponent having had their Territories reduced to zero. 

 

Final Battle 

Creating an End of Game sequence, the Final Battle between Rebel Alliance and 

Federation had to involve the Death Star. Then with the chance of either side winning, I 

had to consider two possible outcomes. The Death Star blown up or the Rebels home 

Planet and last refuge totally destroyed. 

 

 

 

 

 

 

 

 

 

 

 

 

Declared Winner 

To accomplish this is just a printed 

declaration blazed across the screen. 

óRebel Alliance or Federation Wins 

the Gameô and displaying the relative 

Starship.   

 

 

 

League Table 

Having met the challenge and acquired a final victory, itôs not unexpected to want to 

match you results against others. I therefore added a League Table to satisfy this Gamer 

prerogative. QBAD2370DLT Loads at initialisation and with a new high scorer the 

program allows updates to be made and then Saved automatically. 

. 

QBITS Galaxy AD2370 Testing 

Some Galaxy AD2370 checks can be made with the test code built into the program.  

 

 (F1) checks the screen sequences, a Highperspace Jump, Star Approach and Planetary 

Orbit, followed with a test firing of the selected forces Starships laser weapon. Then after 

a short pause, the Starship leaves orbit. The final battle sequence with the approach of the 

DeathStar to the Rebels Planet now takes place.  

 

Depending on which force is chosen from the menu Federation or Rebel Alliance, either 

the DeathStar discharges its weapon and destroys the planet or the DeathStar itself is 

destroyed.  At this point, the Galaxy map shows the status of all the annexed star 

systems. Then after another delay, switches to show the high score League Table.   

 

(F2) activates the Simulator Mode this disables the human player access and assumes 

their role. The simulation mode now plays against the actions of the Computer player. 

 

The League Table 

As the high score results are held on file, this required some means of resetting to zero 

and checking its function with regarded to updates. To this end a test program is provided 

that can be merged with the existing Galaxy program.  

 

LOAD  QBAD2370v4DL01 and MERGE Test Program QBAD2370DLTP.  RUN the 

program and select (N) from Game Menu. Following initialisation press CTRL -

Spacebar to halt the program. Type TP at QL Command line and press ENTER this 

should reset the League Table entries to all zeros and clear any names. To check run 

Galaxy AD2370 again and select (H) from Menu. 
 

Further checks - as supplied QBAD2370v4DLTP has a number of lines REMarked out. 

By removing the REMark on certain lines, you can set new values and force an entry on 

any of the different Skill levels of the League Table.   
 

 

LName Checks skill level and if new high score allows name entry.   

LSave and LLoad   These Save / Load the QBAD2370DLT high score file.  

LScore    Displays the League Table and current entries.  

 

GTS Game Time Session 

GSk=  Skill  Level (1) Marshal (2) Admiral (3) Captain   

GEnd Runs End Game sequence. 
 

Note: Skill Level, Game Time Session, MODs, Territories and Gamer Name are 

contained within file QBAD2370v4DLT. Saved after any update and Loaded before 

Initialisation of Game. 

QBIT Galaxy ADA2017 PROCedures 
 

Intro  Introduction screen to the Game   

InitWin  Initialise the screens (WINDOWS 3 to 18) 

GMark  Galaxy Mark draws Rebel Alliance icon 

SS01/2/3 StarShip image SS01 / SS02 / SS03 

MODL /R MODL - Left image /MODR - Right image 

RollDice Dice throw - random numbers 1 to 8  

DiceA/B Square image / Diamond image 

GMenu Choice of Side /Skill  / (N)ew (L )oad (S)ave (H)ighscore (E)xit 

GSkill  Get Skill Level  

GMPrn  Display Menu choices 

GExit Close unused channels clear screen...  

GNewInit  Initialise the Star / Planet DATA 

InitMap  Initialise the Galaxy Map Grid & Star Systems 

GLoad/GSave Save /Load Game Data file  

SelPath/FChk Select File QBAD2370DL_1-8, File checks on device [flp1_] 

GNotes Galaxy AD2370 Stratagems / Notes 

GameAD2370 Main Game Loop 

StarMove PAN / SCROLL of screens 3 to 10 (Space View) 

PMap Displays Game Players Star System Grid location on Galaxy Map   

CMap Displays Computer Players Star System on Galaxy Map 

HJump HyperSpace Jump [Warp 12] to Star System 

StarView/ Star Star Approach [Warp 1] - Enlarging Star  

Eorbit  Takes Star Ships in to Planetary Orbit (Impulse) 

Lorbit  Leaves Orbit.  

GPlay Player decisions over Attack/ Transfer ï Tec/Trade/MODs etc. 

MODInfo  MODs distribution of Rebel Alliance / Federation Starships &Planets 

PTInfo Planet :- Pop/Tec/Trade/ MODs 

SSInfo  StarShip:- Tec/Trade/ MODs 

Phaser Fires Phasers down on to Planet  

Victory  Updates Attacking Sides Tec/Trade Credits and MODs 

TranPS Transfer Tec/Trade/MODs  Planet >Starship 

TranSP  Transfer Tec/Trade/MODs  Starship>Planet  

Recruit Re-Calculate Pop & MODs after change in Tec/Trade credits  

GEnd Game End Winner sequence ï returns to Game Menu 

DeathStar  DeathStar arrival at a Planet location 

DSAttack Deathstar destroys Planet 

DSDefeat Deathstar is itself destroyed 

LMap  League Table Rebel Alliance or Federation annexed Star Systems 

LScore Displays the League Table of high scores     

LName/ LSave/ LLoad   New Name entry, Save & Load  QBAD2370DLT   

CPlay Computer Player Attack/ Transfer Tec/Trade/MODs 

SPlay Simulated Player Attack/ Transfer Tec/Trade/MODs 
 

Note: SPlay and CPlay can play against each other to check strength of Stratagems. 

QBITS Galaxy AD2370 Flow Chart 

100 REMark QBAD2370v4DL01 - QBITS Galxay AD2370 v4.01 2017        PART1  
 
102 DIM Dat$(8,12)                :REMark Game Data Files 1 to 8 
103 FOR df=1 TO 8:Dat$(df)='QBAD2370DL_'&df 
104 Dn$='flp1_':df=1              :REMark Devicename [Change to User preference] 
105 m=0:slk=1:file=1              :REMark Game Data file New/Load/Save checks 
106 LTFile$=Dn$&ôQBAD2370DLT'    :REMark League Table file 
 
108 DIM astro$(70,20),astro(70,9) :REMark Star/Planet 1-70 DATA 
109 DIM side(2,3),Sclk$(20)       :REMark Game Players 1-2 DATA : GTS h:m:s 
110 DIM name$(4,10),score(3,3)    :REMark HighScore League Table 
111 DIM Gm$(10,70):RESTORE 124:FOR l=1 TO 10:READ Gm$(l) 
 
113 Gp=0:Gp1=0:Gp2=0:xsd=54:sd$='(R)'   :REMark Game Player checks 
114 RMOD=0:RSSn=1:Rt=10          :REMark Rebel Alliance MODs:SShips:Territories 
115 FMOD=0:FSSn=1:Ft=10           :REMark Federation     MODs:SShips:Territories 
116 en=0:s=1:t=1:zr=1:Lck=0      :REMark RollDice:Attack/Transfer:(R)ecruit:LName 
117 dm=3:ws=1:Mes$='          '   :REMark delete MODs:Warp Speed:Messages 
118 GTS=0:Tm1=60:Tm2=20:Test=0   :REMark Game Time :Test=0/1 for checks 
 
120 Mode 4:Intro:GWinInit:GMenu     :REMark Title - Game Menu 
 
122 DATA 7,90,50,7,76,60,4,48,78,4,54,88,4,64,98,4,70,108,4,96,118 
123 DATA 7,130,156,7,160,171,2,188,194 
124 DATA 'It is a turbulent time in the Galaxy as the Federation' 
125 DATA 'seeks to enforce new rules on Interstellar Travel and Trade.' 
126 DATA 'Select a Star System with ¾ ¿ then SpaceBar to HyperSpace Jump to its' 
127 DATA 'location and enter into Planetary Orbit. If Planet already annexed' 
128 DATA 'or taken by your forces, Transfer Tec/Trade credits to increase' 
129 DATA 'the Population. Recruit more MODs to be transported to other' 
130 DATA 'Star Systems and their Planets to Attack or Defend.' 
131 DATA 'Choose Skill Level and which Side to be on' 
132 DATA 'Federation or Rebel Alliance','(M)enu - press any key' 
 
134 DEFine PROCedure Intro 
135 WINDOW#1,512,256,0,0:PAPER#1,0:CLS#1:PAPER#2,0:CLS#2:SCALE#1,150,0,-10 
136 FOR i=1 TO 250:INK RND(2 TO 7):POINT RND(0 TO 230),RND(0 TO 140) 
137 CSIZE#1,2,1:OVER#1,-1 
138 FOR i=2 TO 4:INK#1,i*2:CURSOR#1,138+i,18+i:PRINT#1,'QBITS GALAXY AD2370' 
139 CSIZE#1,0,0:OVER#1,0:RESTORE 122 
140 FOR a=1 TO 2 
141   READ i,c,r:INK i:CURSOR c,r:FOR b=1 TO 70:PAUSE 1:PRINT Gm$(a,b); 
142 END FOR a 
143 n=20:astro$(n)='Sol':astro(n,5)=1:astro(n,8)=243:Gp=1:x=140:y=80:m=3 
144 Star:DeathStar:ink1=7:ink2=241:x=186:y=26:SS02:x=30:y=22:SS03 
145 WINDOW#2,496,220,8,10:PAPER#2,0:INK#2,7:CSIZE#2,0,0:PAUSE 50 
146 x=83:y=40:LINE#2,x-40,y TO x,y+1 TO x+40,y TO x,y-1 TO x-40,y 
147 FILL#2,1:CIRCLE#2,x,y+1,5,.3,PI/2:FILL#2,0 
148 FOR a=3 TO 10:READ i,c,r :INK i:CURSOR c,r :PRINT Gm$(a):PAUSE 5 
149 PAUSE :CURSOR#2,0,116:FOR a=1 TO 16:SCROLL#2,-8,1:SCROLL#2,5,2:PAUSE 3 
150 END DEFine 

Galaxy Symbol   

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
152 DATA 132,50,104,26,90,60,211,26,132,50,276,26  :REMark win 3-10 View screen 
153 DATA 132,40,104,66,132,40,276,66 
154 DATA 132,50,104,98,90,60,211,86,132,50,276,98 
155 DATA 90,84,8,28,90,84,414,28,272,66,120,148    :REMark win 11-13 Starship / Planet /Galaxy Map  
156 DATA 90,22,8,6,90,22,414,6,198,12,214,12       :REMark win 14-16 Star x,y / Name / Message 
157 DATA 90,42,8,130,90,42,414,130                 :REMark win 17-18 Side Status 
 
159 DEFine PROCedure GWinInit 
160 file=1:Gp1=1:Gp2=2:LLoad:RESTORE 152 
161 OPEN#3,scr_496x20a8x4:CLS#3:CSIZE#3,1,1:OVER#3,-1 
162 FOR i=1 TO 2:INK#3,i*2:CURSOR#3,96+i,0:PRINT#3,'GALAXY AD2370' 
163 OVER#3,0:CLOSE#3 
164 FOR i=3 TO 18:OPEN#i,scr_:READ a,b,c,d:WINDOW#i,a,b,c,d 
165 WINDOW#1,312,130,100,24:BORDER#1,2,2:PAPER#1,0:CLS#1 
166 WINDOW#1,300,120,106,26:WINDOW#0,496,20,8,230 
167 BORDER#11,1,2:WINDOW#11,86,74,10,34 
168 BORDER#12,1,2:WINDOW#12,86,74,416,34 
169 BORDER#13,1,7:PAPER#13,0:CLS#13 
170 INK#2,7:CIRCLE#2, 30,21,1.25:INK#2,4:CIRCLE#2, 30,21,3 
171 INK#2,7:CIRCLE#2,136,21,1.25:INK#2,2:CIRCLE#2,136,21,3 
172 MODL:MODR:l=2.9:dx=30:DiceA:dx=136:DiceA 
173 sn=n:StarView:Eorbit:x=30:y=30:SS03 
174 END DEFine 

Game Menu 

This is where you choose the Skill  level to play (1)Marshal - (2)Admiral - (3)Captain and 

decide which Side (R)ebel Alliance or (F)ederation to play on. (S)ave at this point is 

disabled but activated following a (N)ew Game initialisation or (L )oad of a previously 

saved one.  
 

(H)ighscore ï QBAD2370DLT  the current League Table file is loaded at start up. The 

Winners GTS (Game Time Session) is shown with Total MODs, Territories annexed and 

the name, nickname or pseudonym (up to 9 characters) of the current holder.  

 
176 DEFine PROCedure GSkill 
177 IF Gp1=1:xsd=54:sd$='(R)':ELSE xsd=174:sd$='(F)' 
178 IF Tm1=20:Tm$='Marshal' :GSk=1:xsk= 54:sk$='(1)' 
179 IF Tm1=40:Tm$='Admiral' :GSk=2:xsk=120:sk$='(2)' 
180 IF Tm1=60:Tm$='Captain' :GSk=3:xsk=186:sk$='(3)' 
181 INK#2,4:AT#2,21,0:PRINT#2,'Skill Level: ';Tm$ 
182 AT#2,21,63:PRINT#2,'Game Time: ' 
183 END DEFine 
 
185 DEFine PROCedure GMPrn 
186 ch=13:CSIZE#ch,0,0:INK#ch,4 
187 CURSOR#ch,18, 14:PRINT#ch,'Skill (1)Marshal (2)Admiral (3)Captain' 
188 CURSOR#ch,18,38:PRINT#ch,'Side  (R)ebel Alliance or (F)ederation' 
189 INK#ch,7:CURSOR#ch,18,26:PRINT#ch,'(N)ew (L)oad (S)ave (H)ighscore (E)xit' 
190 CURSOR#ch,xsk,13:PRINT#ch,sk$:CURSOR#ch,xsd,37:PRINT#ch,sd$ 
191 END DEFine 
 
193 DEFine PROCedure GMenu 
194 ch=13:CLS#13:CSIZE#ch,2,0:INK#ch,7:OVER#ch,1 
195 FOR i=1 TO 2:CURSOR#ch,78+i,2:PRINT#ch,'Game Menu' 
196 OVER#ch,0:GSkill:GMPrn 
197 REPeat lp 
198  k=CODE(INKEY$(-1)) 
199  SELect ON k 
200   =49:Tm1=20:GSkill:GMPrn           :REMark Marshal 
201   =50:Tm1=40:GSkill:GMPrn           :REMark Admiral 
202   =51:Tm1=60:GSkill:GMPrn           :REMark Captain 
203   =70,102:Gp1=2:Gp2=1:GSkill:GMPrn  :REMark Federation 
204   =82,114:Gp1=1:Gp2=2:GSkill:GMPrn  :REMark Rebel Alliance 
205   =78,110:slk=0:m=1:GNewInit        :REMark (N)ew 
206   =76,108:SelPath  :GLoad           :REMark (L)oad 
207   =83,115:IF slk=0:SelPath:GSave    :REMark (S)ave 
208   =72,104:LScore:PAUSE:GMenu        :REMark (H)ighscore 
209   =69,101:GExit                      :REMark (E)xit 
210   END SELect 
211 END REPeat lp 
212 END DEFine 
 
214 DEFine PROCedure GExit 
215 CLOSE#99:FOR w=3 TO 18:CLOSE#w 
216 INK#2,7:INK#1,7:WINDOW#1,512,256,0,0:PAPER#1,0:CLS#1 
217 INK#0,7:WINDOW#1,490,220,16,8:PRINT#0,'bye...':STOP 
218 END DEFine 

A New Game requires seeding the Galaxy Map with Star Systems and their details: 

Population, Tec/Trade credits and MODs to defend themselves.  
 

Creating a name list of stars required  poring over several star charts and then taking a 

quite liberal use of names, altering a few and making up one or two others. The DATA 

distribution is deliberate so as not to end up in alphabetic order or anything like.  
 
 
220 DATA 'Vega','Bellatrix','Zibal','Castor','Naos','Artos','Ilad' 
221 DATA 'Tabit','Diphda','Elnath','Acrux','Errai','Cellus','Detrus' 
222 DATA 'Polaris','Homan','Rigel','Sabik','Sarin','Westron','Zenda' 
223 DATA 'Furud','Gienah','Hadar','Propus','Izar','Centra','Nebtron' 
224 DATA 'Enif','Algol','Mirzarm','Libeta','Atria','Panus','Ulan' 
225 DATA 'Keid','Kochab','Lesath','Ankaa','Marsic','Nebus','Alean' 
226 DATA 'Meissa','Subra','Menkib','Altair','Muscida','Pusan','Curson' 
227 DATA 'Cursa','Nashira','Nunki','Ogma','Pollux','Xilon','Scarb' 
228 DATA 'Sirius','Merack','Serena','Denab','Tureis','Noran','Theron' 
229 DATA 'Beid','Wesen','Yildun','Caph','Zooma','Jarron','Cetrus' 
 
 
 

(N)ew Initialising Game 
 
 
 
231 DEFine PROCedure GNewInit 
232 RESTORE 219:ch=13:CURSOR#ch,12,50:PRINT#ch,'Initilising';:CLS#ch,4 
233 FOR n=1 TO 70 
234    READ a$:astro$(n)=a$:CURSOR#ch,80+n*2,50:PRINT#ch,'.';:PAUSE 1 
235    pop=RND(2 TO 10)/RND(2 TO 8):IF pop<.5:pop=.5 
236    astro(n,1)=INT(pop*1.2)      :REMark Tec 
237    astro(n,2)=INT(pop*2)        :REMark Trade 
238    astro(n,3)=INT(pop*2)        :REMark MODs 
239    astro(n,4)=pop               :REMark Pop 
240    astro(n,5)=0                 :REMark Side 
241    astro(n,6)=n*4+15            :REMark x coordinate 
242    astro(n,7)=RND(10 TO 90)     :REMark y coordinate 
243    astro(n,8)=RND(220 TO 254)   :REMark Planet Colour 
244    astro(n,9)=RND(1 TO 3)       :REMark Orbit Approach 
245 END FOR n 
246 REMark astro(n,5) 0=Independent 1=Rebel Alliance 2=Federation 
247 FOR i=1 TO 10 
248    n=RND(15 TO 55):IF astro(n,5)=0:astro(n,5)=2:ELSE GO TO 247 
249 END FOR i 
250 FOR i=1 TO  5 
251    n=RND( 5 TO 20):IF astro(n,5)=0:astro(n,5)=1:ELSE GO TO 250 
252    n=RND(50 TO 65):IF astro(n,5)=0:astro(n,5)=1:ELSE GO TO 251 
253 END FOR i 
254 FOR n=1 TO 70:IF astro(n,5)>0 AND astro(n,3)<3:astro(n,3)=3 
255 FOR Gp=1 TO 2 :side(Gp,1)=3:side(Gp,2)=5:side(Gp,3)=0 
256 GMapInit:GTS=0:GAD2370 
257 END DEFine 

259 DEFine PROCedure GMapInit 
260 ch=13:CLS#ch:BORDER#ch,1,7:INK#ch,0,2,3:SCALE#ch,100,0,0 
261 FOR h=12 TO 96 STEP 18 
262  LINE#ch,6,h TO 310,h 
263 END FOR h 
264 FOR v=16 TO 330 STEP 20 
265  LINE#ch,v,3 TO v,98 
266 END FOR v 
267 INK#ch,2:FOR i=1 TO 60 :POINT#ch,RND(60 TO 260),RND(30 TO 70) 
268 INK#ch,4:FOR i=1 TO 30 :POINT#ch,RND(80 TO 240),RND(40 TO 60) 
269 INK#ch,7:FOR i=1 TO 180:POINT#ch,RND(10 TO 300),RND(10 TO 90) 
270 ch=16:CLS#ch:INK#ch,4 
271 PRINT#ch,'Use ¾¿    to Select Star System':BLOCK#ch,16,4,40,4,4 
272 ch=1:CLS#ch:INK#ch,7:FOR i=1 TO 120:POINT#ch,RND(10 TO 200),RND( 5 TO 95) 
273 END DEFine 
 
 
 
 
 
 
 
 
 
275 DEFine PROCedure SelPath 
276 ch=13:CSIZE#ch,0,0:INK#ch,2:CURSOR#ch,0,50:CLS#ch,4 
277 CURSOR#ch,180,50:PRINT#ch,'        <Enter/Esc>':INK#ch,7 
278 REPeat Path_lp 
279  CURSOR#ch,12,50:PRINT#ch,'Select: ';Dn$&Dat$(df) 
280  k=CODE(INKEY$(-1)) 
281  SELect ON k 
282   =208:df=df+1:IF df>8:df=1 
283   =216:df=df-1:IF df<1:df=8 
284   = 10:file=1:device_filename$=Dn$&Dat$(df):EXIT Path_lp 
285   = 27:file=0:EXIT Path_lp 
286  END SELect 
287 END REPeat Path_lp 
288 device_filename$=Dn$&Dat$(df) 
289 END DEFine 
 
 
  
 SelPath selects filename for Save/ Load 
 

 
 
 
 
 

 FChck returns Not Found if not previously saved. 

291 DEFine PROCedure FChck 
292 ch=13:CURSOR#ch,12,50:PRINT#ch,'Searching...':CLS#ch,4 
293 DELETE Dn$&'FList' 
294 ch=99:OPEN_NEW#ch,Dn$&'FList':DIR#ch,Dn$:CLOSE#ch 
295 OPEN_IN#ch,Dn$&'FList' 
296 REPeat Dir_lp 
297   IF EOF(#ch) 
298      CLOSE#ch:ch=13:CURSOR#ch,12,50:PRINT#ch,'File Not Found...' 
299      PAUSE 50:CURSOR#ch,0,50:CLS#ch,4:file=0:slk=1:RETurn 
300   END IF 
301   INPUT#ch,Fchk$:IF Fchk$==Dat$(df):CLOSE#ch:EXIT Dir_lp 
302 END REPeat Dir_lp 
303 ch=13:CURSOR#ch,12,50:PRINT#ch,'Loading':CLS#ch,4 
304 END DEFine 
 
306 DEFine PROCedure GSave 
307 ch=13:IF file=0:CURSOR#ch,0,50:CLS#ch,4:RETurn 
308 DELETE device_filename$ 
309 CURSOR#ch,12,50:PRINT#ch,'Saving':CLS#ch,2:CLS#ch,4 
310 OPEN_NEW#99,device_filename$ 
311 FOR a=1 TO 70 
312   CURSOR#ch,66+a*2,50:PRINT#ch,'.';:PRINT#99,astro$(a):PAUSE 1 
313   FOR b=1 TO 9:PRINT#99,astro(a,b):END FOR b 
314 END FOR a 
315 FOR a=1 TO 2 
316   FOR b=1 TO 3:PRINT#99,side(a,b) :END FOR b 
317 END FOR a 
318 PRINT#99,GTS\Tm1\Gp1\Gp2:CLOSE#99:CURSOR#ch,0,40:CLS#ch,2 
319 END DEFine  

 
 

(S)ave Game Data to file 
 
 
 

(L)oad Game Data File 
 
 
321 DEFine PROCedure GLoad 
322 ch=13:IF file=0:CURSOR#ch,0,50:CLS#ch,4:RETurn 
323 FChck:IF file=0:CURSOR#ch,0,50:CLS#ch,4:RETurn 
324 OPEN_IN#99,device_filename$ 
325 FOR a=1 TO 70 
326   CURSOR#ch,72+a*2,50:PRINT#ch,'.';:INPUT#99,astro$(a):PAUSE 1 
327   FOR b=1 TO 9:INPUT#99,astro(a,b):END FOR b 
328 END FOR a 
329 FOR a=1 TO 2 
330   FOR b=1 TO 3:INPUT#99,side(a,b) :END FOR b 
331 END FOR a 
332 INPUT#99,GTS\Tm1\Gp1\Gp2:CLOSE#99:CURSOR#ch,0,40:CLS#ch,2 
333 slk=0:GSkill:GMapInit:GAD2370 
334 END DEFine 

QBITS Galaxy AD2370 Starships & MODS 

A number of ideas to incorporate Starship images was part of my agenda. Creating these 

can be done in several ways, with a pixel by pixel grid, generated from a scanned image 

or simply by use of SuperBASIC commands ARC, CIRCLE, LINE, POINT. That is not 

forgetting the accompanying, INK, FILL, OVER, PI, and use of SCALE.  

 

Creating a Starship 

The range of designs are as you would expect numerous and can get quite complex, my 

aim was to keep it as simple as possible. Starting with a few doodles to identify a usable 

image, it was then broken down into possible sections. As a child I remember being busy 

for hours linking up sets of points to create a picture. With a little imagination, 

employing this technique an image can be broken down into a number of shapes using 

LINE and CIRCLE commands. Then the image can be build from overlapping layers.   
 
 

 

 

 
The one shown has 5 overlapping 

shapes, a filled quadrilateral for 

the far wing, a filled white circle 

for the main body with a black ink 

circle drawn over the top as an 

outline. A black filled circle for 

the flight deck. A 2nd quadrilateral 

for the nearside wing.  

 

Individual elements of an image are coded separately within their own PROCedure 

where the elements of LINE, CIRCLE etc. are grouped together with colour settings and 

x, y off-sets to global coordinates.  

 

Being able to set the ink colours deployed in the 

image Procedure allows the useful possibility to 

redraw with all colours set to the background, so it 

can in affect be made to disappear.    

 

Creating the MODs 

The LINE values again are straightforward to work out. 
 

 

 

 

 

 

 

if  
 

The other shapes can be generated in much the same way see opposite page -


